9

REPRESENTACIÓN GRÁFICA DE SUPERFICIES

9.1.Figuras Predefinidas

▼ Función Graphics3D[ ]

★ Esferas

De centro {1, 1, 1} con radio 2

```
Graphics3D[Sphere[{0, 0, 0}, 2]]
```



Varias esferas

```
Graphics3D[{Sphere[{0, 0, 0}, 1], Sphere[{3, 3, 0}, 2]}]
```



★ Cilindros

```
Graphics3D[Cylinder[]]
```



Cilindro desde el origen hasta el punto {1, 1, 1} con radio 2

```
Graphics3D[{Opacity[0.8], Cylinder[{{0, 0, 0}, {1, 1, 1}}, 2]},
 Axes → True, AxesLabel → {"x", "y", "z"}]
```



★ Conos

```
Graphics3D[Cone[]]
```



```
Graphics3D[Cone[{{0, 0, 0}, {1, 1, 1}}, 1 / 2]]
```



9.2.Curvas y Superficies Parametrizadas

▼ Función ParametricPlot3D[ ]

★ ParametricPlot3D[{funciónx ,funcióny, funciónz},{u, umin, umáx}]

Con esta función se representa una curva parametrizada en el espacio, donde la variable u varia de umin a umax

```
c = ParametricPlot3D[{Sin[u], Cos[u], u / 10},
  {u, 0, 20}, PlotStyle → Directive[Darker[Green], Thickness[0.01]]]
```



★ ParametricPlot3D[{funciónx ,funcióny, funciónz},{u, umin, umáx},{v, vmin, vmáx}]

Con esta función se representa una superficies parametrizada en el espacio, donde las variables u y v varian de umin a umax y de vmin a vmax

```
s = ParametricPlot3D[{v * Sin[u], v * Cos[u], u / 10}, {u, 0, 4 Pi}, {v, 0, 1},
  PlotStyle → Directive[Purple, Opacity[0.5]], BoxRatios → Automatic, Mesh → 5]
```




```
Show[{s, c}, Axes → False]
```



▼ Opciones de ParametricPlot

```
ParametricPlot3D[{Cos[ϕ] Sin[θ], Sin[ϕ] Sin[θ], Cos[θ]},
 {ϕ, 0, 2 π}, {θ, 0, π}, PlotPoints → 100, Mesh → None,
 ColorFunction → Function[{x, y, z, ϕ, θ}, Hue[Sin[6 ϕ] Sin[6 θ]]],
 ColorFunctionScaling → False]
```



▼ Coordenadas esféricas

Podemos utilizar las coordenadas esfericas para la representación de una superficie:

x(u,v)= r Sin[u] Sin[v]

y(u,v)= r Cos[u] Sin[v]

z(u,v)= r Cos[v]

★ Esfera

```
ParametricPlot3D[ {2 Sin[u] Sin[v], 2 Cos[u] Sin[v], 2 Cos[v]},
 {u, -π, π}, {v, -π, π}, PlotStyle -> Opacity[0.5]]
```



▼ Coordenadas Cilíndricas

Podemos utilizar las coordenadas cilíndricas para la representación de una superficie:

x(u,v)= r Cos[u]

y(u,v)= r Sin[u]

z(u,v)= v

★ Cilindro

Un cilindro es la cincunferencia de radio 2 desde z=0 hasta z=1

```
ParametricPlot3D[{2 Cos[u], 2 Sin[u], v}, {u, 0, 2 Pi}, {v, 0, 1},
 Mesh → 5, BoundaryStyle → Black, PlotStyle → FaceForm[Pink, Blue]]
```



★ Cono

```
ParametricPlot3D[{v Cos[u], v Sin[u], 2 v}, {u, 0, 2 Pi}, {v, 0, 1},
 Mesh → 5, BoundaryStyle → Black, PlotStyle → FaceForm[Purple, Green]]
```



★ Elipsoide

```
ParametricPlot3D[{2 Cos[u] Sin[v], Sin[u] Sin[v], Cos[v]}, {v, 0, Pi}, {u, 0, 2 Pi},
 Mesh → 5, BoundaryStyle → Black, PlotStyle → FaceForm[Orange, Yellow]]
```



★ Toro de Revolución

```
ParametricPlot3D[{(2 + Cos[v]) Cos[u], (2 + Cos[v]) Sin[u], Sin[v]}, {u, 0, 2 Pi},
 {v, 0, 2 Pi}, Mesh → 5, BoundaryStyle → Black, PlotStyle → FaceForm[Purple, Green]]
```



★ Banda de Möbius

```
ParametricPlot3D[{Cos[t] (3 + r Cos[t / 2]), Sin[t] (3 + r Cos[t / 2]), r Sin[t / 2]},
 {r, -1, 1}, {t, 0, 2 Pi}, Mesh → {5, 10}, PlotStyle → FaceForm[Orange, Yellow]]
```



9.3.Regiones en el Espacio

▼ Función RegionPlot3D[ ]

Esta función permite dibujar el interior del paraboloide

```
RegionPlot3D[x^2 + y^2 < z, {x, -2, 2}, {y, -2, 2}, {z, 0, 2}]
```



```
RegionPlot3D[x^2 + y^2 < z, {x, 0, 2}, {y, -2, 2}, {z, 0, 2}]
```



▼ Función ContourPlot3D[ ]

Esta función devuelve la superficie del paraboloide

```
ContourPlot3D[x^2 + y^2 == z, {x, -2, 2}, {y, -2, 2}, {z, 0, 2}]
```



★ Cubo

```
RegionPlot3D[-1 ≤ x ≤ 1 && -1 ≤ y ≤ 1 && -1 ≤ z ≤ 1,
 {x, -2, 2}, {y, -2, 2}, {z, -2, 2}, Mesh → None, PlotPoints → 50]
```



★ Corteza esférica

```
RegionPlot3D[1 ≤ x^2 + y^2 + z^2 ≤ 3, {x, -2, 2},
 {y, -2, 2}, {z, -2, 0}, Mesh → None, PlotPoints → 50]
```



★ Otras regiones

```
RegionPlot3D[x^2 + y^2 + z^2 ≤ 4 && y^2 + z^2 ≤ 3, {x, -2, 2}, {y, -2, 2}, {z, -2, 2}]
```



```
RegionPlot3D[x^2 + z^2 ≤ 4 && y^2 + z^2 ≤ 3, {x, -2, 2}, {y, -2, 2}, {z, -2, 2}]
```



```
Plot3D[10 - x^2 - 2 y^2, {x, -3, 3}, {y, -3, 3},
 RegionFunction → Function[{x, y, z}, 8 < 2 x^2 + 3 y^2 < 10]]
```



9.4.Superficies de Revolución

★ **Astroide**

```
ParametricPlot[{Cos[t]^3, Sin[t]^3}, {t, 0, 2 π}, AspectRatio → Automatic,
 PlotStyle → {RGBColor[0.7, 0.3, 0.2], Thickness[0.01]},
 PlotLabel → Style["ASTROIDE", Bold, Blue, 14]]
```



```
RevolutionPlot3D[{Cos[t]^3, Sin[t]^3}, {t, -Pi / 2, Pi / 2}, AspectRatio → Automatic]
```



★ Paraboloide de Revolución

```
ParametricPlot[{t, 0.5 t^2}, {t, -4, 4}, AxesOrigin → {0, 0},
 AspectRatio → Automatic, PlotStyle → {RGBColor[0.7, 0.3, 0.2], Thickness[0.01]},
 PlotLabel → Style["PARÁBOLA", Bold, Blue, 14]]
```



```
RevolutionPlot3D[{t, 0.5 t^2}, {t, 0, 2 Pi}]
```



★ Toro de Revolución

```
ParametricPlot[{{-2 - Cos[t], Sin[t]}, {2 + Cos[t], Sin[t]}},
 {t, 0, 2 π}, PlotRange -> {{-3.1, 3.1}, {-1.5, 1.5}},
 AxesOrigin → {0, 0}, AspectRatio → Automatic, PlotStyle →
  {{RGBColor[0.7, 0.3, 0.2], Thickness[0.01]}, {RGBColor[0.7, 0.3, 0.2], Dashing[0.01]}},
 PlotLabel → Style["CIRCUNFERENCIA de centro (0,2) y radio 1", Bold, Blue, 14]]
```



```
ParametricPlot3D[{(2 + Cos[v]) Cos[u], (2 + Cos[v]) Sin[u], Sin[v]}, {u, 0, 2 Pi},
 {v, 0, 2 Pi}, Mesh → 5, BoundaryStyle → Black, PlotStyle → FaceForm[Purple, Green]]
```